BOX ... FILL COLOR

PURPOSE
Specifies the color to use for the interior region of a box to appear on subsequent plots (as specified by the BOX ... COORDINATES command) or boxes generated by the BOX command. Up to 100 boxes can be specified by the BOX ... COORDINATES command. See the Color chapter for a list of available colors.

DESCRIPTION
DATAPLOT can draw either solid filled boxes or hatch pattern filled boxes. This command is used in either case. The BOX FILL PATTERN command is used to set the type of pattern fill.

SYNTAX
BOX <id> FILL COLOR <color>
where <id> is an integer number or parameter in the range 1 to 100 that specifies the box identifier; and <color> is a string or an integer number or parameter that specifies the desired color.

EXAMPLES
BOX 2 FILL COLOR BLACK
BOX 1 FILL COLOR RED
BOX 1 FILL COLOR 6

NOTE 1
Boxes can be generated in two ways. You can use the BOX COORDINATES command to place boxes on subsequent plots or you can use the BOX command to draw one box immediately. The BOX FILL COLOR command is used to specify the color of the box pattern fill in either case. When the BOX command is used, the color is set to the line style of the first box (i.e., <id> is 1).

NOTE 2
A BOX ... FILL COLOR command with no arguments sets the color to default. Thus BOX 1 FILL COLOR with no arguments sets the fill color for box 1 to default. BOX FILL COLOR command with no <id> refers to all 100 boxes; thus BOX FILL COLOR RED assigns the color red to all 100 boxes. BOX FILL COLOR with no <id> and no arguments sets the default fill color for all 100 boxes.

DEFAULT
Box interiors are drawn in black for both solid fills and hatch pattern fills.

SYNONYMS
None

RELATED COMMANDS
PLOT = Generates a data or function plot.
BOX COORDINATES = Generate a box on subsequent plots.
BOX = Draws a box (immediately).
BOX COLOR = Sets the colors for box borders.
BOX PATTERN = Sets the line styles for box borders.
BOX THICKNESS = Sets the line thicknesses for box borders.
BOX FILL GAP = Sets the spacings for box hatch patterns.
BOX FILL LINE = Sets the line types for box hatch patterns.
BOX FILL PATTERN = Sets the fill patterns for boxes.
BOX FILL THICKNESS = Sets the line thicknesses for box hatch patterns.
BAR PATTERN COLOR = Sets the colors for bar pattern fills.
BAR FILL COLOR = Sets the colors for bar solid fills.

APPLICATIONS
Presentation Graphics

IMPLEMENTATION DATE
92/8
PROGRAM
LET CARTER = DATA 66 30 11 43 44 41 35 82 54 36
LET REAGAN = DATA 26 54 84 47 51 51 52 14 36 55
LET X = DATA 1 2 3 5 6 7 8 10 11 12; LET X1 = X - 0.2; LET X2 = X + 0.2
.
YLIMITS 1 12; YTIC OFFSET 1 1
MINOR YTIC MARK NUMBER 0
YTIC LABEL CONTENT DEMOCRATS INDEPENDENTS REPUBLICANS SP() EAST ...
SOUTH MIDWEST WEST SP() BLACKS HISPANICS WHITES SP()
XLABEL PERCENT; XLIMITS 0 100
MAJOR XTIC NUMBER 11; MINOR XTIC NUMBER 1
.
LINE BLANK BLANK; BAR ON ON
BAR WIDTH 0.2 ALL; BAR DIRECTION H ALL
BAR FILL ON ON
BAR COLOR G25 G75
BOX FILL PATTERN ON
BOX SHADOW HW 0 0
BOX 1 CORNER COORDINATES 70 80 73 83
BOX 1 FILL COLOR G25
BOX 2 CORNER COORDINATES 70 85 73 88
BOX 2 FILL COLOR G75
LEGEND JUSTIFICATION LECE
LEGEND 1 CARTER; LEGEND 1 COORDINATES 74 81.5
LEGEND 2 REAGAN; LEGEND 2 COORDINATES 74 86.5
TITLE DEMONSTRATE BOX FILL COLOR
TITLE SIZE 5
PLOT X1 CARTER AND
PLOT X2 REAGAN

[Diagram of box fill color demonstration]

DEMONSTRATE BOX FILL COLOR

- Whites
- Hispanics
- Blacks
- West
- Midwest
- South
- East
- Republicans
- Independents
- Democrats

PERCENT

CARTER

REAGAN